

Each tab demonstrates a specific Tables capability. Work through in order. Each tab includes a before/after comparison showing the plain range approach vs the Table approach.

TAB 1 — CONVERT AND NAME: UAT TEST LOG

Scenario: A plain range test log with 200 rows. Status column used by COUNTIFS formulas on a Summary sheet. New scenarios are added weekly.

Before: plain range COUNTIFS

Formula

```
=COUNTIFS(C:C,"Willow",E:E,"Fail")
```

When 50 new rows are added at row 251, all COUNTIFS formulas miss them. The Summary shows wrong totals. No warning is given.

Exercise: Convert range to Table named TestLog. Rewrite all COUNTIFS formulas to use TestLog[Module] and TestLog[Status]. Add 10 rows. Verify the Summary updates.

After: structured reference COUNTIFS

Formula

```
=COUNTIFS(TestLog[Module],"Willow",TestLog[Status],"Fail")
```

New rows are included automatically. Column names are self-documenting. If Module moves to a different column, the formula still works.

Key result: same count, but this formula is correct in perpetuity regardless of Table size.

TAB 2 — CALCULATED COLUMNS: DENIAL PRIORITY FLAGS

Scenario: ClaimData Table with columns: ClaimID, Payer, ServiceType, Status, DenialCode, Amount, SubmitDate. Add three calculated columns that populate automatically for every row.

Calculated Column 1: DenialFlag

Formula

```
=IF([@Status]="Denied","Flag", "")
```

Written once in the first cell of a blank column. Excel fills to every row in ClaimData automatically.

Exercise: Delete the formula from rows 5-10. Notice Excel restores it immediately when you tab away — calculated columns enforce consistency.

Calculated Column 2: DaysOpen

Formula

```
=NETWORKDAYS([@SubmitDate],TODAY())
```

Business days since submission. Calculates for every row automatically. Add a new claim row — DaysOpen is calculated immediately.

Exercise: Add 5 rows below the last row. Verify DaysOpen is populated without any action.

Calculated Column 3: Escalate

Formula

```
=IF(AND([@Status]="Denied",OR([@Payer]="BCBS",[@Payer]="UHC"),[@Amount]>EscalationThreshold),"Escalate", "")
```

EscalationThreshold is a named range from a Config sheet. When the threshold changes, all Escalate values update on next calculation.

Exercise: Change EscalationThreshold from 1000 to 500. Verify all Escalate flags update to reflect the lower threshold.

TAB 3 — POWER QUERY SOURCE: MONTHLY CLAIM REFRESH

Scenario: ClaimRaw Table (the paste target) connects to a Power Query that standardises and categorises claim data, loading results to ClaimClean Table. Pivot tables source from ClaimClean.

Monthly workflow — plain range version (the problem)

Formula

Paste new data. Update Power Query source range. Update all pivot table source ranges. Verify COUNTIFS range covers new rows. Fix any formulas that missed new rows.

Total time: 30-90 minutes per month depending on workbook complexity.

Risk: Missed range updates produce wrong report numbers silently.

Monthly workflow — Table version (the solution)

Formula

Paste new data immediately below ClaimRaw Table header row. Press Ctrl+Alt+F5 (Refresh All).

Total time: under 3 minutes. All Power Query transformations and pivot tables update automatically.

Exercise: Simulate by pasting 20 rows into ClaimRaw. Press Refresh All. Verify ClaimClean Table and all pivot tables reflect the new rows.

TAB 4 — XLOOKUP WITH TABLES: MIGRATION RECONCILIATION

Scenario: Two Tables: SourceCounts (entity type and count from legacy system extract) and TargetCounts (entity type and count from Epic target). Reconciliation Table uses XLOOKUP to compare.

Reconciliation formula — static range version (the problem)

Formula

```
=XLOOKUP(A2,$F$2:$F$15,$G$2:$G$15,0)
```

If a new entity type is added to SourceCounts at row 16, the formula range \$F\$2:\$F\$15 misses it.

Fix requires manually updating the range in every XLOOKUP formula.

Reconciliation formula — Table version (the solution)

Formula

```
=XLOOKUP([@Entity],SourceCounts[Entity],SourceCounts[Count],0)
```

SourceCounts Table auto-expands when new entity types are added. XLOOKUP always searches the full Table.

Exercise: Add "Allergies" to SourceCounts and TargetCounts. Verify the Reconciliation Table picks up the new entity type and calculates variance correctly without formula edits.